
Supplementary material for “Bayesian

joint ordinal and survival modeling for

breast cancer risk assessment”

C. Armero(1), C. Forné(2,3), M. Rué(2,4), A. Forte(1), H. Perpiñán(1,5), G.

Gómez(6), and M. Baré(4,7)

In this supplemental document, we provide additional data analysis results and the BUGS code of the proposed

model for studying breast cancer diagnosis with regard to longitudinal breast density measurement.

BI-RADS and age

Table A1. Posterior expectation of the probability associated to each BI-RADS category with respect to age.

Age (years)
BI-RADS 50 55 60 65 70

a 0.150 0.168 0.191 0.214 0.239
b 0.526 0.538 0.541 0.545 0.545
c 0.166 0.157 0.147 0.136 0.125
d 0.158 0.137 0.121 0.106 0.092

Cumulative incidence

In Figure A1, the dashed line shows the cumulative incidence calculated using the Kaplan-Meier approach without

accounting for competing risk events. The solid line shows the cumulative incidence, after adjusting for competing

risk. The dotted line shows the cumulative incidence of the death occurring prior to the breast cancer diagnosis.

Figure A1.The estimated cumulative incidence of breast cancer using the Kaplan-Meier approach and the competing risks approach.

JAGS code for the proposed model

model{

for(i in 1:N){

for(j in 1:K[i]){ #1

y[i,j] ~ dcat(p[i,j,1:4])

p[i,j,1] <- 1-q[i,j,1]

p[i,j,2] <- q[i,j,1]-q[i,j,2]

p[i,j,3] <- q[i,j,2]-q[i,j,3]

p[i,j,4] <- q[i,j,3]

logit(q[i,j,1]) <- location[i,j]-gamma1

logit(q[i,j,2]) <- location[i,j]

logit(q[i,j,3]) <- location[i,j]-gamma3

location[i,j] <- beta0 + b0[i] + beta1*time[i,Proj[i,j]]

+ b1[i]*time[i,Proj[i,j]]

}#times[i,] a grid of times for individual i and times[i,Proj[i,j]]

is the observed time of individual i at time j

#Ones’ trick for a not implemented likelihood in BUGS

ones[i] ~ dbern(prob[i]) #2

prob[i] <- L[i]/C

## Likelihood for survival data - event=1 if event, and 0 if censoring

L[i] <- (pow(f[i],event[i])*pow(S[i],1-event[i]))/(S0[i])

## Hazard for individual i at each time point of the grid time[i,]

for(j in 1:J[i]){

h[i,j] <- (lambda*pow(time[i,j], (lambda-1))) * exp(eta[1] + eta[2]*numrel[i]

+ eta[3]*brstproc[i] + alpha*(beta0 + b0[i] + (beta1 + b1[i])*time[i,j]))

}

## Cumulative hazard H[t] = int_0^t (h[u] du) using GAUSS-LEGENDRE polynomials

and evaluated at 12 different (n.end) points indicated by Proj.aux

for(j in 1:n.end){ #3

Aux[i,j] <- ((time[i,Proj.aux[j+1]]-time[i,Proj.aux[j]])/2)

*inprod(wk.int, h[i, (Proj.aux[j]+1):(Proj.aux[j]+5)])

#wk.int Gauss-Legendre weights

H[i,j] <- sum(Aux[i,1:j])

}

## Survival probability for individual i at their final time

S[i] <- exp(-H[i,n.end])

## Survival probability for individual i at their entry time

S0[i] <- exp(-H[i,n.ini])

## Density function

f[i] <- h[i,J[i]]*S[i]

b0[i] ~ dnorm(0, tau0)

b1[i] ~ dnorm(0, tau1)

}

C <- 10000

#Prior distributions for the survival process

for(i in 1:3){ eta[i] ~ dnorm(0.0, 0.0001) }

alpha ~ dnorm(0.0, 0.0001)

lambda ~ dgamma(1, 1)

#Prior distributions for the longitudinal process

beta0 ~ dnorm(0.0, 0.0001)

beta1 ~ dnorm(0.0, 0.0001)

gamma1 ~ dnorm(-1.0986, 0.001)I(,0)

gamma3 ~ dnorm(1.0986, 0.001)I(0,)

tau0 <- pow(sig0, -2)

sig0 ~ dunif(0, 100)

tau1 <- pow(sig1, -2)

sig1 ~ dunif(0, 100)

}

1. Specification of the latent variable

2. One’s trick for the Survival likelihood to incorporate left truncation.

3. Code for the numerical integration for H(t) using Gauss-Legendre polynomials and evaluated at different

time points

